Example 8.12 Distribution of Overpotential in a Porous Electrode
Equation (8.1.28) is solved in Maple and the results obtained are given below:

```
> restart : with(inttrans) : with(plots) :
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2)-nu^2*u(x,t);
```

$$eq := \frac{\partial}{\partial t}\, u(x,t) = \frac{\partial^2}{\partial x^2}\, u(x,t) - \nu^2\, u(x,t) \tag{1}$$

```
> u(x,0):=0;
```

$$u(x,0) := 0 \tag{2}$$

```
> bc1:=diff(u(x,t),x)=-delta;
```

$$bc1 := \frac{\partial}{\partial x}\, u(x,t) = -\delta \tag{3}$$

```
> bc2:=diff(u(x,t),x)=delta*beta;
```

$$bc2 := \frac{\partial}{\partial x}\, u(x,t) = \delta\,\beta \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\,U(x) = \frac{d^2}{dx^2}\, U(x) - \nu^2\, U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx}\, U(x) = -\frac{\delta}{s} \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := \frac{d}{dx}\, U(x) = \frac{\delta\,\beta}{s} \tag{7}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1 \sin\left(\sqrt{-s-\nu^2}\; x\right) + \_C2 \cos\left(\sqrt{-s-\nu^2}\; x\right) \tag{8}$$

```
> U(x):=c[1]*cosh((s+nu^2)^(1/2)*x)+c[2]*sinh((s+nu^2)^(1/2)*x);
```

$$U(x) := c_1 \cosh\left(\sqrt{s+\nu^2}\; x\right) + c_2 \sinh\left(\sqrt{s+\nu^2}\; x\right) \tag{9}$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{\delta\left(\cosh\left(\sqrt{s+\nu^2}\right) + \beta\right) \cosh\left(\sqrt{s+\nu^2}\; x\right)}{\sinh\left(\sqrt{s+\nu^2}\right)\sqrt{s+\nu^2}\; s} - \frac{\delta \sinh\left(\sqrt{s+\nu^2}\; x\right)}{\sqrt{s+\nu^2}\; s} \tag{10}$$

```
> U(x):=factor(combine(simplify(U(x))));
```

$$U(x) := \frac{\delta\left(\cosh\left(\sqrt{s+\nu^2}\,(x-1)\right)+\cosh\left(\sqrt{s+\nu^2}\,x\right)\beta\right)}{\sinh\left(\sqrt{s+\nu^2}\right)\sqrt{s+\nu^2}\,s} \quad \textbf{(11)}$$

```
> U1(x):=subs(s=s-nu^2,U(x));
```

$$U1(x) := \frac{\delta\left(\cosh\left(\sqrt{s}\,(x-1)\right)+\cosh\left(\sqrt{s}\,x\right)\beta\right)}{\sinh\left(\sqrt{s}\right)\sqrt{s}\,\left(s-\nu^2\right)} \quad \textbf{(12)}$$

```
> P(s):=numer(U1(x));
```

$$P(s) := -\delta\left(\cosh\left(\sqrt{s}\,(x-1)\right)+\cosh\left(\sqrt{s}\,x\right)\beta\right) \quad \textbf{(13)}$$

```
> Q(s):=denom(U1(x));
```

$$Q(s) := \sinh\left(\sqrt{s}\right)\sqrt{s}\,\left(-s+\nu^2\right) \quad \textbf{(14)}$$

```
> solve(Q(s),s);
```

$$0,\ \nu^2 \quad \textbf{(15)}$$

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \quad \textbf{(16)}$$

```
> solve(Q(s),s);
```

$$-\pi^2\,\_Z1{\sim}^2,\ 0,\ \nu^2 \quad \textbf{(17)}$$

The roots are:

```
> 0,0,nu^2,-n^2*Pi^2;
```

$$0,\ 0,\ \nu^2,\ -n^2\pi^2 \quad \textbf{(18)}$$

```
> mu0:=0;
```

$$\mu 0 := 0 \quad \textbf{(19)}$$

```
> b[2]:=(s-mu0)^2*P(s)/Q(s);
```

$$b_2 := -\frac{s^{3/2}\,\delta\left(\cosh\left(\sqrt{s}\,(x-1)\right)+\cosh\left(\sqrt{s}\,x\right)\beta\right)}{\sinh\left(\sqrt{s}\right)\left(-s+\nu^2\right)} \quad \textbf{(20)}$$

```
> B[2]:=limit(b[2],s=0);
```

$$B_2 := 0 \quad \textbf{(21)}$$

```
> b[1]:=diff(b[2],s):
> B[1]:=limit(b[1],s=0);
```

$$B_1 := \frac{-\delta\beta-\delta}{\nu^2} \quad \textbf{(22)}$$

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

**(23)**

$$A_n := \frac{2\,\delta\left(\cosh\left(\sqrt{\mu}\,(x-1)\right)+\cosh\left(\sqrt{\mu}\,x\right)\beta\right)\sqrt{\mu}}{\cosh\left(\sqrt{\mu}\right)\mu^{3/2}-\cosh\left(\sqrt{\mu}\right)\sqrt{\mu}\,\nu^2+3\sinh\left(\sqrt{\mu}\right)\mu-\sinh\left(\sqrt{\mu}\right)\nu^2} \tag{23}$$

```
> A[0]:=subs(mu^(1/2)=nu,mu^(3/2)=nu^3,mu=nu^2,A[n]):
> A[0]:=simplify(A[0]);
```

$$A_0 := \frac{\delta\left(\cosh\left(\nu\,(x-1)\right)+\cosh(\nu\,x)\,\beta\right)}{\nu\sinh(\nu)} \tag{24}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*n*Pi,mu^(3/2)=-I*n^3*Pi^3,mu=-
n^2*Pi^2,A[n])):
> vars:={cos(n*Pi)=(-1)^n,sin(n*Pi)=0};
```

$$vars := \left\{\cos(n\,\pi)=(-1)^n,\ \sin(n\,\pi)=0\right\} \tag{25}$$

```
> A[n]:=simplify(subs(vars,A[n])):
> A[n]:=simplify(subs(vars,expand(A[n])));
```

$$A_n := -\frac{2\,\delta\cos(n\,\pi\,x)\left(1+(-1)^{-n}\beta\right)}{n^2\pi^2+\nu^2} \tag{26}$$

```
> b1s:=B[1]*subs(mu0=0,1/(s-mu0));
```

$$b1s := \frac{-\delta\,\beta-\delta}{\nu^2\,s} \tag{27}$$

```
> b1t:=invlaplace(b1s,s,t);
```

$$b1t := -\frac{(1+\beta)\,\delta}{\nu^2} \tag{28}$$

```
> b2s:=B[2]*subs(mu0=0,1/(s-mu0)^2);
```

$$b2s := 0 \tag{29}$$

```
> b2t:=invlaplace(b2s,s,t);
```

$$b2t := 0 \tag{30}$$

```
> u0s:=subs(mu=nu^2,A[0]/(s-mu));
```

$$u0s := \frac{\delta\left(\cosh\left(\nu\,(x-1)\right)+\cosh(\nu\,x)\,\beta\right)}{\nu\sinh(\nu)\left(s-\nu^2\right)} \tag{31}$$

```
> u0t:=invlaplace(u0s,s,t);
```

$$u0t := \frac{\delta\left(\cosh\left(\nu\,(x-1)\right)+\cosh(\nu\,x)\,\beta\right)e^{\nu^2 t}}{\nu\sinh(\nu)} \tag{32}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := -\frac{2\,\delta\cos(n\,\pi\,x)\left(1+(-1)^{-n}\beta\right)}{\left(n^2\pi^2+\nu^2\right)(s-\mu)} \tag{33}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := -\frac{2\,\delta\cos(n\,\pi\,x)\,\left(1+(-1)^{-n}\,\beta\right)\,\mathrm{e}^{\mu\,t}}{n^2\,\pi^2+\nu^2} \tag{34}$$

```
> unt:=subs(mu=-n^2*Pi^2,unt);
```

$$unt := -\frac{2\,\delta\cos(n\,\pi\,x)\,\left(1+(-1)^{-n}\,\beta\right)\,\mathrm{e}^{-n^2\,\pi^2\,t}}{n^2\,\pi^2+\nu^2} \tag{35}$$

```
> U:=b1t*exp(-nu^2*t)+b2t*exp(-nu^2*t)+simplify(u0t*exp(-nu^2*t))+
  exp(-nu^2*t)*Sum(unt,n=1..infinity);
```

$$U := -\frac{(1+\beta)\,\delta\,\mathrm{e}^{-\nu^2 t}}{\nu^2}+\frac{\delta\,\left(\cosh(\nu\,(x-1))+\cosh(\nu\,x)\,\beta\right)}{\nu\sinh(\nu)}+\mathrm{e}^{-\nu^2 t}\left(\sum_{n=1}^{\infty}\left(-\frac{2\,\delta\cos(n\,\pi\,x)\,\left(1+(-1)^{-n}\,\beta\right)\,\mathrm{e}^{-n^2\,\pi^2\,t}}{n^2\,\pi^2+\nu^2}\right)\right) \tag{36}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
> pars:={nu=1,delta=1,beta=0.1};
```

$$pars := \{\nu = 1,\ \beta = 0.1,\ \delta = 1\} \tag{37}$$

```
> plot3d(subs(pars,u),x=0..1,t=0..0.5,axes=boxed,title="Figure
  Exp. 8.21.",labels=[x,t,"u"],orientation=[-60,60]);
```

Figure Exp. 8.21.




```
> plot([subs(t=0,pars,u),subs(t=0.05,pars,u),subs(t=0.1,pars,u),
  subs(t=0.2,pars,u)],x=0..1,axes=boxed,title="Figure Exp. 8.22.",
  thickness=5,labels=[x,"u"]);
```

Figure Exp. 8.22.




In all the examples discussed in this chapter until now, the roots of q(s) were obtained analytically. This is not always possible. Often the roots should be obtained numerically as in section 7.1.4. This is illustrated in the next example.

>